

## CHAPTER

## 1

## Getting Started with the Output Delivery System

|   |    |
|---|----|
| <i>What Is the Output Delivery System?</i>                  | 3  |
| <i>Creating Listing Output</i>                              | 4  |
| <i>Creating Printer Output</i>                              | 4  |
| <i>Creating HTML Output</i>                                 | 6  |
| <i>Identifying Output Objects</i>                           | 8  |
| <i>Selecting Output Objects to Send to ODS Destinations</i> | 10 |
| <i>Creating an Output Data Set</i>                          | 12 |
| <i>Storing Links in the Results Folder</i>                  | 14 |
| <i>Customizing Procedure Results</i>                        | 16 |

### What Is the Output Delivery System?

Prior to Version 7, SAS procedures that produced printed output (that is, output destined for the procedure output file) generated output that was designed for a traditional line-printer. This type of output has limitations that prevent users from getting the most value from their results:

- Traditional SAS output is limited to monospace fonts. In this day of desktop document editors and publishing systems, users want more versatility in printed output.
- Traditional SAS output provides no way for you to parse its contents. You can't, for example, know in advance in what column the values for the third variable in a report begin.
- Quite a few commonly used procedures did not produce output data sets. Users who wanted to use output from one of these procedures as input to another procedure relied on PROC PRINTTO and the DATA step to retrieve results that could not be stored in an output data set.

Beginning with Version 7, procedure output became much more flexible. The Output Delivery System (ODS) has been designed to overcome the limitations of traditional SAS output and to make it easy to make new formatting options available to users. ODS is a method of delivering output in a variety of formats and of making the formatted output easy to access. Important features of ODS include the following:

- ODS combines raw data with one or more table definitions to produce one or more output objects. These objects can be sent to any or all ODS destinations. The currently available ODS destinations can produce an output data set, traditional monospace output, output that is formatted for a high-resolution printer, and output that is formatted in Hypertext Markup Language (HTML).

- ODS provides table definitions that define the structure of the output from procedures and from the DATA step. You can customize the output by modifying these definitions or creating your own.
- ODS provides a way for you to choose individual output objects to send to ODS destinations. For instance, PROC UNIVARIATE produces five output objects. You can easily create HTML output, an output data set, traditional Listing output, or Printer output from any or all of these output objects. You can send different output objects to different destinations.
- ODS stores a link to each output object in the Results folder in the Results window.

In addition, ODS removes responsibility for formatting output from individual procedures and from the DATA step. The procedure or DATA step supplies raw data and the name of the table definition that contains the formatting instructions, and ODS formats the output. Because formatting is now centralized in ODS, the addition of a new ODS destination does not affect any procedures or the DATA step. As future destinations are added to ODS, they will automatically become available to all procedures that support ODS and to the DATA step.

This document provides several different kinds of information about ODS:

- This section shows you how to accomplish some of the basic tasks that you are most likely to want to do with the Output Delivery System: how to send procedure results to each of the ODS destinations and how to customize the appearance of procedure output. More detailed information and examples appear in the reference sections of this document.
- Chapter 2, “Basic Concepts of the Output Delivery System,” on page 21 discusses the concepts that will help you to understand and to use ODS to your best advantage.
- Chapter 3, “The ODS Statements,” on page 47 provides reference information on the ODS statements. These statements control the many features of the Output Delivery System.
- Chapter 4, “Using the Output Delivery System in the DATA Step,” on page 105 explains how to use ODS with the DATA step.
- Chapter 5, “The TEMPLATE Procedure,” on page 131 explains how to customize your output.

---

## Creating Listing Output

You do not need to change your SAS programs to create Listing output. By default, the Listing destination is open. Unless you specifically close the Listing destination with the ODS LISTING CLOSE statement, you will continue to create Listing output.

---

## Creating Printer Output

If you open the Printer destination, you can create output that is formatted for a high-resolution printer. The first ODS PRINTER statement in the following SAS program opens the Printer destination and directs the formatted output to the file `odsprinter.ps`. The second ODS PRINTER statement closes the Printer destination. You must close the Printer destination before you can print the file.

The data set STATEPOP is created in a DATA step on page 73. The REGFMT format is created in a PROC FORMAT step on page 81. The Printer output appears in Display 1.1 on page 5.

*Note:* This example uses file names that may not be valid in all operating environments. To successfully run the example in your operating environment, you may need to change the file specifications. See Appendix 1, “Alternative ODS HTML Statements for Running Examples in Different Operating Environments,” on page 275.  $\Delta$

```
options nodate nonumber;
ods printer file='odsprinter.ps';
proc tabulate data=statepop;
  class region state;
  var citypop_80 citypop_90;
  table region*state, citypop_80*sum=' ' citypop_90*sum=' ';
  format region regfmt.;
  where region=1;
  label citypop_80='1980' citypop_90='1990';
  title 'Metropolitan Population for the Northeast Region';
  title2 '(measured in millions)';
run;
ods printer close;
```

**Display 1.1** Output Created by the Printer Destination

***Metropolitan Population for the Northeast Region***  
***(measured in millions)***

|                   |       | 1980  | 1990  |
|-------------------|-------|-------|-------|
| Geographic region | State |       |       |
| Northeast         | CT    | 2.98  | 3.15  |
|                   | MA    | 5.53  | 5.79  |
|                   | ME    | 0.41  | 0.44  |
|                   | NH    | 0.54  | 0.66  |
|                   | NJ    | 7.37  | 7.73  |
|                   | NY    | 16.14 | 16.52 |
|                   | PA    | 10.07 | 10.08 |
|                   | RI    | 0.89  | 0.94  |
|                   | VT    | 0.13  | 0.15  |

For more information about creating Printer output, see “ODS PRINTER Statement” on page 65.

*Note:* Procedure options that affect presentation may not affect Printer output. For instance, the DOUBLE option in PROC PRINT, which inserts a blank line between observations, has no effect on Printer output.  $\Delta$

---

## Creating HTML Output

If you open the HTML destination, you can create output that is formatted in Hypertext Markup Language (HTML). You can browse these files with Internet Explorer, Netscape, or any other browser that fully supports the HTML 3.2 tag set.

The ODS HTML statement, which generates the HTML files, can create

- an HTML file (called the body file) that contains the results from the procedure
- a table of contents that links to the body file
- a table of pages that links to the body file
- a frame that displays the table of contents, the table of pages, and the body file.

For example, the first ODS HTML statement in the following SAS program generates four HTML files. ODS routes the results of the PROC UNIVARIATE step to the body file as well as to the Listing destination. ODS also creates the associated contents, page, and frame files. The second ODS HTML statement closes the HTML destination. You must close the HTML destination before you can browse the HTML files.

*Note:* This example uses file names that may not be valid in all operating environments. To successfully run the example in your operating environment, you may need to change the file specifications. See Appendix 1, “Alternative ODS HTML Statements for Running Examples in Different Operating Environments,” on page 275.  $\Delta$

```
/* Create HTML files. */
ods html file='odshtml-body.htm'
         contents='odshtml-contents.htm'
         page='odshtml-page.htm'
         frame='odshtml-frame.htm';

proc univariate data=statepop mu0=3.5;
  var citypop_90 noncitypop_90;
  title;
run;

/* Close the HTML destination.          */
ods html close;
```

The frame file appears in Display 1.2 on page 7.

Display 1.2 First View of the Frame File

| <i>The UNIVARIATE Procedure</i>                                 |            |                  |            |
|---|------------|------------------|------------|
| <i>Variable: CityPop_90 (1990 metropolitan pop in millions)</i> |            |                  |            |
| Moments   |            |                  |            |
| N   | 51         | Sum Weights      | 51         |
| Mean  | 3.87701961 | Sum Observations | 197.728    |
| Std Deviation   | 5.16465302 | Variance         | 26.6736408 |
| Skewness  | 2.87109259 | Kurtosis         | 10.537867  |
| Uncorrected SS  | 2100.27737 | Corrected SS     | 1333.68204 |
| Coeff Variation   | 133.21194  | Std Error Mean   | 0.72319608 |

| Basic Statistical Measures |          |                     |          |
|----------------------------|----------|---------------------|----------|
| Location                   |          | Variability         |          |
| Mean                       | 3.877020 | Std Deviation       | 5.16465  |
| Median                     | 2.423000 | Variance            | 26.67364 |
| Mode                       | .        | Range               | 28.66500 |
|                            |          | Interquartile Range | 3.60000  |

**Table of Contents**

- The Univariate Procedure
  - CityPop\_90
    - Moments
    - Basic Measures of Location and Variability
    - Tests For Location
    - Quantiles
    - Extreme Observations
  - NonCityPop\_90
    - Moments
    - Basic Measures of Location and Variability
    - Tests For Location
    - Quantiles
    - Extreme Observations
    - Missing Values

**Table of Pages**

- The Univariate Procedure
  - Page 1
  - Page 2

If you click on **Extreme Observations** under **CityPop\_90** in the Table of Contents, the HTML table that contains that part of the procedure results appears at the top of the frame that contains the body file (see Display 1.3 on page 7).

Display 1.3 Second View of the Frame File

| Extreme Observations |     |         |     |
|----------------------|-----|---------|-----|
| Lowest               |     | Highest |     |
| Value                | Obs | Value   | Obs |
| 0.134                | 41  | 10.083  | 9   |
| 0.152                | 3   | 12.023  | 18  |
| 0.191                | 39  | 14.166  | 26  |
| 0.221                | 36  | 16.515  | 7   |
| 0.226                | 50  | 28.799  | 49  |

*The UNIVARIATE Procedure*

*Variable: NonCityPop\_90 (1990 nonmetropolitan pop in million)*

| Moments       |            |                  |            |
|---------------|------------|------------------|------------|
| N             | 49         | Sum Weights      | 49         |
| Mean          | 1.04042857 | Sum Observations | 50.981     |
| Std Deviation | 0.66036104 | Variance         | 0.43607671 |
| Skewness      | 0.46479456 | Kurtosis         | -0.4491065 |
| Uncorrected   | 73.02271   | Corrected SS     | 30.02160   |

**Table of Contents**

- The Univariate Procedure
  - CityPop\_90
    - Moments
    - Basic Measures of Location and Variability
    - Tests For Location
    - Quantiles
    - Extreme Observations
  - NonCityPop\_90
    - Moments
    - Basic Measures of Location and Variability
    - Tests For Location
    - Quantiles
    - Extreme Observations
    - Missing Values

**Table of Pages**

- The Univariate Procedure
  - Page 1
  - Page 2

For more information about creating HTML output, see “ODS HTML Statement” on page 50. You can see many examples of HTML output in *SAS Procedures Guide* online documentation.

*Note:* Procedure options that affect presentation may not affect HTML output. For instance, the `DOUBLE` option in `PROC PRINT`, which inserts a blank line between observations, has no effect on HTML output.  $\Delta$

---

## Identifying Output Objects

Several of the features of the Output Delivery System (such as selecting and excluding output objects to send to ODS destinations or creating a data set from an output object) require that you specify one or more particular output objects for the system to process. You identify an output object by its name, its label, or its path. To learn what these are, run your SAS program preceded by this statement:

```
ods trace on;
```

As long as the tracing feature is on, information about each output object that is created appears in the SAS log.

Use this statement to stop sending the information to the log:

```
ods trace off;
```

For example, the following SAS program produces the SAS log that is shown in Output 1.1 on page 9:

```
options nodate pageno=1 linesize=64 pagesize=60;
ods trace on;

proc univariate data=statepop mu0=3.5;
  var citypop_90 noncitypop_90;
  title;
run;

ods trace off;
```

**Output 1.1** SAS Log Produced by the ODS TRACE Statement

Compare the second output object that is created for CityPop\_90 to the second output object that is created for NonCityPop\_90. These objects are marked with an arrow (<---). The names and labels of these objects are identical. Thus, using a name or a label can refer to multiple output objects, which is sometimes useful. If you want to reference each output object separately, you must use its path, which is unique.

```

36  options nodate pageno=1 linesize=64 pagesize=60;
37  ods trace on;
38
39  proc univariate data=statepop mu0=3.5;
40      var citypop_90 noncitypop_90;
41      title;
42  run;

Output Added:
-----
Name:      Moments
Label:     Moments
Template:  base.univariate.Moments
Path:     Univariate.CityPop_90.Moments
-----

Output Added:
-----
Name:      BasicMeasures <---
Label:     Basic Measures of Location and Variability
Template:  base.univariate.Measures
Path:     Univariate.CityPop_90.BasicMeasures
-----

Output Added:
-----
Name:      TestsForLocation
Label:     Tests For Location
Template:  base.univariate.Location
Path:     Univariate.CityPop_90.TestsForLocation
-----

Output Added:
-----
Name:      Quantiles
Label:     Quantiles
Template:  base.univariate.Quantiles
Path:     Univariate.CityPop_90.Quantiles
-----

Output Added:
-----
Name:      ExtremeObs
Label:     Extreme Observations
Template:  base.univariate.ExtObs
Path:     Univariate.CityPop_90.ExtremeObs
-----

```

```

Output Added:
-----
Name:      Moments
Label:     Moments
Template:  base.univariate.Moments
Path:     Univariate.NonCityPop_90.Moments
-----

Output Added:
-----
Name:      BasicMeasures <---
Label:     Basic Measures of Location and Variability
Template:  base.univariate.Measures
Path:     Univariate.NonCityPop_90.BasicMeasures
-----

Output Added:
-----
Name:      TestsForLocation
Label:     Tests For Location
Template:  base.univariate.Location
Path:     Univariate.NonCityPop_90.TestsForLocation
-----

Output Added:
-----
Name:      Quantiles
Label:     Quantiles
Template:  base.univariate.Quantiles
Path:     Univariate.NonCityPop_90.Quantiles
-----

Output Added:
-----
Name:      ExtremeObs
Label:     Extreme Observations
Template:  base.univariate.ExtObs
Path:     Univariate.NonCityPop_90.ExtremeObs
-----

Output Added:
-----
Name:      MissingValues
Label:     Missing Values
Template:  base.univariate.Missings
Path:     Univariate.NonCityPop_90.MissingValues
-----

```

If you compare this SAS log to the Results Folder that appears in Display 1.6 on page 15, you can see that the string that identifies the output in the Results folder is its label.

For more information about the trace record, see “ODS TRACE Statement” on page 71.

---

## Selecting Output Objects to Send to ODS Destinations

Some procedures, such as PROC UNIVARIATE, produce multiple output objects. Any procedure that uses ODS produces multiple output objects when you use BY-group processing. ODS enables you to select which of these output objects go to the open ODS destinations. ODS destinations include the Listing destination, the HTML destination, the Printer destination, and the Output destination. For more information on ODS destinations, see “ODS Destinations” on page 23.

You choose the objects to send to destinations with the ODS SELECT or the ODS EXCLUDE statement. To select individual output objects, use this form of the ODS SELECT statement:

```
ODS SELECT selection(s);
```

where each *selection* can be a full path, a name, or a label (see the trace record in Output 1.1 on page 9). You can also use a partial path. A partial path consists of any part of the full path that begins immediately after a period (.) and continues to the end of the full path. For details on referencing output objects, see “How to Specify an Output Object” on page 24.

For example, to select just the output objects that contain the basic measures and the quantiles from the PROC UNIVARIATE output, use the following program.

*Note:* This example uses file names that may not be valid in all operating environments. To successfully run the example in your operating environment, you may need to change the file specifications. See Appendix 1, “Alternative ODS HTML Statements for Running Examples in Different Operating Environments,” on page 275.  $\Delta$

```
/* Create HTML files. */
ods html body='select-body.htm'
      contents='select-contents.htm'
      page='select-pages.htm'
      frame='select-frame.htm';

/* Select output objects by name. */
ods select BasicMeasures Quantiles;

/* Analyze the data.      */
proc univariate data=statepop mu0=3.5;
  var citypop_90 noncitypop_90;
  title;
run;

/* Close the HTML destination. */
ods html close;
```

The frame file appears in Display 1.4 on page 12. The program also creates Listing output, which is not shown. The Listing output contains the same information as the HTML body file, but it is formatted with the traditional SAS monospace font.

**Display 1.4** View of the Frame File for Selected Output Objects

The contents file shows that for each variable in the analysis, PROC UNIVARIATE produces two output objects: one that contains basic measures and one that contains quantiles. All four output objects are in the body file because the ODS SELECT statement used names to identify the objects. If the ODS SELECT statement had used paths, which are unique, it could have selected output objects for the individual variables.

| Basic Statistical Measures |          |                     |          |
|----------------------------|----------|---------------------|----------|
| Location                   |          | Variability         |          |
| Mean                       | 3.877020 | Std Deviation       | 5.16465  |
| Median                     | 2.423000 | Variance            | 26.67364 |
| Mode                       | .        | Range               | 28.66500 |
|                            |          | Interquartile Range | 3.60000  |

| Quantiles (Definition 5) |          |
|--------------------------|----------|
| Quantile                 | Estimate |
| 100% Max                 | 28.799   |
| 99%                      | 28.799   |
| 95%                      | 14.166   |
| 90%                      | 9.574    |
| 75% Q3                   | 4.376    |
| 50% Med                  | 2.423    |
| 25% Q1                   | 0.776    |
| 10%                      | 0.000    |

For more information on selecting output objects, see “ODS SELECT Statement” on page 68.

## Creating an Output Data Set

The Output Delivery System enables you to create a data set from an output object. To create a data set, use the ODS OUTPUT statement. In this statement, you identify

- one or more output objects from which to create a data set
- the names of the data sets to create.

To create a single output data set, use this simplified form of the ODS OUTPUT statement:

ODS OUTPUT *output-object*=*SAS-data-set*;

Specify the output object as you do in the ODS SELECT statement: with a path, a name, a label, or a partial path. For example, to generate and print an output data set from each output object that contains the basic measures that PROC UNIVARIATE produces, use the following SAS program.

*Note:* This example uses file names that may not be valid in all operating environments. To successfully run the example in your operating environment, you may need to change the file specifications. See Appendix 1, “Alternative ODS HTML Statements for Running Examples in Different Operating Environments,” on page 275.  $\triangle$

```

/* Turn off the generation of Listing output */
/* because you want to create a data set, not */
/* see the results. */
ods listing close;

/* Specify the data set to create. */
ods output BasicMeasures=measures;

/* When PROC UNIVARIATE runs, ODS */
/* creates a data set named MEASURES */
/* from the output object named */
/* BasicMeasures. */
proc univariate data=statepop mu0=3.5;
  var citypop_90 noncitypop_90;
  title;
run;

/* Open the HTML destination for PROC PRINT. */
ods html body='measures-body.htm'
  contents='measures-contents.htm'
  frame='measures-frame.htm';

/* Print the output data set. */
proc print data=measures noobs headings=horizontal;
  title 'Output Data Set Produced from';
  title2 'PROC UNIVARIATE Basic Measures';
run;

/* Reset the destinations to their defaults. */
/* Close the HTML destination. */
ods html close;
/* Open the Listing destination. */
ods listing;

```

You can use the resulting data set as input to another SAS program. This program simply prints the data set to illustrate its structure. The HTML output from PROC PRINT follows:

**Display 1.5** PROC PRINT Report of the Data Set Created by PROC UNIVARIATE and ODS

The data set contains observations for each of the variables in the VAR statement in PROC UNIVARIATE.

| VarName       | LocMeasure | LocValue | VarMeasure          | VarValue |
|---------------|------------|----------|---------------------|----------|
| CityPop_90    | Mean       | 3.877020 | Std Deviation       | 5.16465  |
| CityPop_90    | Median     | 2.423000 | Variance            | 26.67364 |
| CityPop_90    | Mode       | .        | Range               | 28.66500 |
| CityPop_90    |            | -        | Interquartile Range | 3.60000  |
| NonCityPop_90 | Mean       | 1.040429 | Std Deviation       | 0.66036  |
| NonCityPop_90 | Median     | 0.961000 | Variance            | 0.43608  |
| NonCityPop_90 | Mode       | 0.608000 | Range               | 2.75600  |
| NonCityPop_90 |            | -        | Interquartile Range | 1.12700  |

For more information on creating output data sets, see “ODS OUTPUT Statement” on page 61.

## Storing Links in the Results Folder

When you run a procedure that supports ODS, SAS automatically stores a link to each piece of ODS output in the Results folder in the Results window. It marks the link with an icon that identifies the output destination that created the output.

Consider the following SAS program, which generates Listing, HTML, and Printer output as well as an output data set (Output output). The data set STATEPOP contains information about the distribution of the United States’ population in metropolitan and nonmetropolitan areas for 1980 and 1990. A DATA step on page 73 creates this data set.

```
options nodate pageno=1 linesize=80 pagesize=34;
```

```
ods html file='results-body.htm';
ods printer file='results.ps';
ods output basicmeasures=measures;
```


```
proc univariate data=statepop mu0=3.5;
  var citypop_90 noncitypop_90;
  title;
run;
```

```
ods html close;
ods printer close;
ods output close;
```

The Results folder shows the folders and output objects that the procedure produces.

**Display 1.6** View of the Results Folder

PROC UNIVARIATE generates a folder called Univariate in the Results folder. Within this folder are two more folders: one for each variable in the VAR statement. These folders each contain a folder for each output object. Within the folder for each output object is a link to each piece of output. The icon next to the link indicates which ODS destination created the output. You can see that the Moments object was sent to the Listing, HTML, and Printer destinations while the Basic Measures of Location and Variability was sent to the Listing, HTML, Printer, and Output destinations.



## Customizing Procedure Results

Many procedures that fully support ODS provide table definitions that enable you to customize each output object that the procedure produces. You can do this by creating an alternative table definition for the procedure to use. This section illustrates how to make an alternative table definition. The explanation here focuses on the structure of the table. For detailed explanations of all the statements and attributes that the program uses, see Chapter 5, “The TEMPLATE Procedure,” on page 131.

For example, the following SAS program creates a customized table definition for the BasicMeasures output object from PROC UNIVARIATE. (The trace record provides the name of the table definition that each object uses. See Output 1.1 on page 9.) In the customized version

- the measures of variability precede the measures of location
- the column headers are modified
- statistics are displayed in a bold, italic font with a 7.3 format.

The customized HTML output object appears in Display 1.7 on page 19. The customized Listing output appears in Output 1.2 on page 19. The customized Printer output appears in Display 1.8 on page 20.

*Note:* This example uses file names that may not be valid in all operating environments. To successfully run the example in your operating environment, you may need to change the file specifications. See Appendix 1, “Alternative ODS HTML Statements for Running Examples in Different Operating Environments,” on page 275.  $\Delta$

```

/* These four options all affect the Listing output. */
/* NODATE and NONUMBER also affect the Printer output.*/
/* None of them affects the HTML output.           */
options nodate nonumber linesize=80 pagesize=60;

/* This PROC TEMPLATE step creates a table definition */
/* base.univariate.Measures in the SASUSER template */
/* store. Table definitions that are provided         */
/* by SAS Institute are stored in a template         */
/* store in the SASHELP library. By default, ODS     */
/* searches for a table definition in SASUSER before */
/* SASHELP, so when PROC UNIVARIATE calls for a     */
/* table definition by this name, ODS uses the one   */
/* from SASUSER.                                    */
proc template;
  define table base.univariate.Measures;

  notes "Basic measures of location and variability";

  translate _val_ = ._ into '';

  /* The HEADER statement determines the order */
  /* in which the table definition uses the   */
  /* headers, which are defined later.       */
  header h1 h2 h3;

```

```

/* The COLUMN statement determines the order */
/* in which the variables appear. PROC      */
/* UNIVARIATE names the variables.         */
column VarMeasure VarValue LocMeasure LocValue;

/* These DEFINE blocks define the headers.  */
/* They specify the text for each header. By */
/* default, a header spans all columns, so   */
/* H1 does so. H2 spans the variables       */
/* VarMeasure and VarValue. H3 spans       */
/* LocMeasure and LocValue.                 */
define h1;
  text "Basic Statistical Measures";
  spill_margin=on;
  space=1;
end;

define h2;
  text "Measures of Variability";
  start=VarMeasure;
  end=VarValue;
end;

define h3;
  text "Measures of Location";
  start=LocMeasure;
  end=LocValue;
end;

/* These DEFINE blocks specify characteristics */
/* for each of the variables. There are two   */
/* differences between these DEFINE blocks and */
/* the ones in the table definition in SASHELP. */
/* These blocks use FORMAT= to specify a format */
/* of 7.3 for LocValue and VarValue. They also */
/* use STYLE= to specify a bold, italic font  */
/* for these two variables. The STYLE= option  */
/* does not affect the Listing output.        */
define LocMeasure;
  print_headers=off;
  glue=2;
  space=3;
  style=rowheader;
end;

define LocValue;
  print_headers=off;
  space=5;
  format=7.3;
  style=data{font_style=italic font_weight=bold};
end;

```

```

define VarMeasure;
  print_headers=off;
  glue=2;
  space=3;
  style=rowheader;
end;

define VarValue;
  print_headers=off;
  format=7.3;
  style=data{font_style=italic font_weight=bold};
end;

/* End the table definition. */
end;
/* Run the procedure. */
run;

/* Begin the program that uses the */
/* customized table definition. */

/* The ODS HTML statement opens the HTML */
/* destination and identifies the files to */
/* write to. */
ods html file='statepop-body.htm'
  contents='statepop-contents.htm'
  page='statepop-page.htm'
  frame='statepop-frame.htm';

/* The ODS PRINTER statement opens the */
/* Printer destination and identifies the */
/* file to write to. */
ods printer file='statepop.ps';

/* The ODS SELECT statement selects just the */
/* output object that contains the basic measures. */
ods select BasicMeasures;

/* PROC UNIVARIATE produces one object for each */
/* variable. It uses the customized table */
/* definition to format the data because the */
/* customized definition is in SASUSER. (See the */
/* explanation with the PROC TEMPLATE statement in */
/* this example.) */
title;
proc univariate data=statepop mu0=3.5;
  var citypop_90 noncitypop_90;
run;

/* Close the HTML destination. */

```

```
ods html close;

/* Close the Printer destination. */
ods printer close;
```

**Display 1.7** Customized HTML Output from PROC UNIVARIATE

The screenshot shows the HTML output of PROC UNIVARIATE. On the left is a navigation pane with a 'Table of Contents' and 'Table of Pages' section. The main content area displays two tables of basic statistical measures. The first table is for 'CityPop\_90 (1990 metropolitan pop in millions)' and the second is for 'NonCityPop\_90 (1990 nonmetropolitan pop in million)'. Each table is titled 'Basic Statistical Measures' and is divided into 'Measures of Variability' and 'Measures of Location'.

| Basic Statistical Measures |        |                      |       |
|----------------------------|--------|----------------------|-------|
| Measures of Variability    |        | Measures of Location |       |
| Std Deviation              | 5.165  | Mean                 | 3.877 |
| Variance                   | 26.674 | Median               | 2.423 |
| Range                      | 28.665 | Mode                 | .     |
| Interquartile Range        | 3.600  |                      |       |

| Basic Statistical Measures |       |                      |       |
|----------------------------|-------|----------------------|-------|
| Measures of Variability    |       | Measures of Location |       |
| Std Deviation              | 0.660 | Mean                 | 1.040 |
| Variance                   | 0.436 | Median               | 0.961 |
| Range                      |       | Mode                 | .     |
| Interquartile Range        |       |                      |       |

**Output 1.2** Customized Listing Output from PROC UNIVARIATE

| The UNIVARIATE Procedure                                 |        |                      |       |
|--|--------|----------------------|-------|
| Variable: CityPop_90 (1990 metropolitan pop in millions) |        |                      |       |
| Basic Statistical Measures                               |        |                      |       |
| Measures of Variability                                  |        | Measures of Location |       |
| Std Deviation  | 5.165  | Mean                 | 3.877 |
| Variance   | 26.674 | Median               | 2.423 |
| Range  | 28.665 | Mode                 | .     |
| Interquartile Range                                      | 3.600  |                      |       |

```

The UNIVARIATE Procedure
Variable: NonCityPop_90 (1990 nonmetropolitan pop in million)

Basic Statistical Measures

Measures of Variability          Measures of Location
Std Deviation                    0.660  Mean                    1.040
Variance                        0.436  Median                   0.961
Range                            2.756  Mode                     0.608
Interquartile Range              1.127
    
```

**Display 1.8** Customized Printer Output from PROC UNIVARIATE

*The UNIVARIATE Procedure*  
Variable: *CityPop\_90 (1990 metropolitan pop in millions)*

| Basic Statistical Measures |        |                      |       |
|----------------------------|--------|----------------------|-------|
| Measures of Variability    |        | Measures of Location |       |
| Std Deviation              | 5.165  | Mean                 | 3.877 |
| Variance                   | 26.674 | Median               | 2.423 |
| Range                      | 28.665 | Mode                 | .     |
| Interquartile Range        | 3.600  |                      |       |

*The UNIVARIATE Procedure*  
Variable: *NonCityPop\_90 (1990 nonmetropolitan pop in million)*

| Basic Statistical Measures |       |                      |       |
|----------------------------|-------|----------------------|-------|
| Measures of Variability    |       | Measures of Location |       |
| Std Deviation              | 0.660 | Mean                 | 1.040 |
| Variance                   | 0.436 | Median               | 0.961 |
| Range                      | 2.756 | Mode                 | 0.608 |
| Interquartile Range        | 1.127 |                      |       |

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *The Complete Guide to the SAS® Output Delivery System, Version 8*, Cary, NC: SAS Institute Inc., 1999. 310 pp.

**The Complete Guide to the SAS® Output Delivery System, Version 8**

Copyright © 1999 by SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-425-X

All rights reserved. Produced in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

**U.S. Government Restricted Rights Notice.** Use, duplication, or disclosure of the software and related documentation by the U.S. government is subject to the Agreement with SAS Institute and the restrictions set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries.® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.